
1. Usin g RANDOMIZE with seeds

by Jacques Groslouis
September 20, 2009

The purpose of this document is to describe an observed difference in using the RANDOMIZE function
plus seed with TI Basic and with TI Extended Basic. 

The TI Extended Basic manual1 (page 151) describes the RANDOMIZE function as follows:

�The RANDOMIZE statement resets the random number generator to an unpredictable sequence. If
RANDOMIZE is followed by a numeric-expression, the same sequence of random numbers is produced
each time the statement is executed with that value for the expression. Different values give different
sequences.�

The TI-99/4A User’s Reference Guide2 (page II-95) describes the RANDOMIZE function in a bit more
detail as follows:

�The RANDOMIZE statement is used in conjunction with the random number function (RND). When
the RANDOMIZE statement is not used, the random number function will generate the same sequence
of pseudo-random numbers each time the program is run. When the RANDOMIZE statement is used
without a seed, a different and unpredictable sequence of random numbers is generated by the random
number function each time the program is run. If you use the RANDOMIZE statement with a seed
specified, then the sequence of random numbers generated by the random number function depends upon
the value of the seed. If the same seed is used each time the program is run, then the same sequence of
numbers is generated. If a different seed is used each time the program is run, then a different sequence
of numbers is generated. The seed may be any numeric expression. The number actually used for the seed
is the first two bytes of the internal representation of the number. … Thus, it is possible that the same
sequence of numbers may be generated even if you specify different seeds. For example, RANDOMIZE
1000 and RANDOMIZE 1099 produce the same first two bytes internally and thus the same sequence of
numbers. If the seed you specify is not an integer, then the value used is INT(seed).�

Although these descriptions are similar in describing the results when a program containing the
RANDOMIZE function is run with a seed and then run again with the same seed, its use with TI Extended
Basic produces a peculiar result that is somewhat predictable. 

The observed difference is that with TI Extended Basic the first random fraction generated by any specific
seed from 0 to 99 is built from the same starting number, with only the first two digits in the random
fraction changing. The same thing happens with the second random fraction generated but starting from
a different number. This also happens with the third, fourth and the following random fractions. TI Basic
does not seem to produce the same sequence of random fractions where only the first two digits change
and the remaining digits remain unchanged. The remainder of this document will only deal with what has
been observed with TI Extended Basic.

This table shows the first six random fractions generated using RANDOMIZE seeds of 0 to 6.

Seed RND fraction

1 2 3 4 5 6

0 0.211324865 0.063675343 0.747072754 0.148079781 0.399281999 0.056075622

1 0.421324865 0.473675343 0.357072754 0.958079781 0.409281999 0.266075622

2 0.631324865 0.883675343 0.967072754 0.768079781 0.419281999 0.476075622

3 0.841324865 0.293675343 0.577072754 0.578079781 0.429281999 0.686075622

4 0.051324865 0.703675343 0.187072754 0.388079781 0.439281999 0.896075622

5 0.261324865 0.113675343 0.797072754 0.198079781 0.449281999 0.106075622

6 0.471324865 0.523675343 0.407072754 0.008079781 0.459281999 0.316075622

+.21 +.41 +.61 +.81 +.01 +.21

Note that the random fractions in column 1 increase by .21 and that the other columns increase by .41,
.61, .81 and .01 respectively with the remaining digits from the starting numbers staying the same. If the
number created exceeds 1, then the leading digit is removed. In column 6 a difference of .21 is used again
and the sequence then repeats with the next random fraction using .41, followed by .61, etc. For lack of
a better name, this series of differences has been called "keys." Each column repeats the indicated sequence
up to and including seed 99. Seed 100 generates a series of random fractions that are identical to those
generated by seed 1. Starting with seed 101 the key becomes a four-digit decimal number added to the
same bases, but this has not been investigated any further. 

A list of starting numbers can be generated by creating an Extended Basic program using RANDOMIZE
0 and then repeating PRINT RND the desired number of times.

Except for determining the starting numbers, the above observations can be expressed as a formula that
does not require the use of RANDOMIZE and RND:

Random fraction = MOD (key × 10^12 × seed + starting number, 10^14) / 10^14

Some explanations of the formula: 

The values for the variables key, seed, and starting number should be integers.

MOD ensures that the random fraction, after division by 10^14, does not produce a number greater than
one.  The formula uses the MOD function as it is used in TI Multiplan; TI Extended Basic does not have
a MOD function.

The two-digit key is multiplied by 10^12 to make it a 14-digit number. The keys are 21, 41, 61, 81, and
01.

The starting number is a 14-digit number and must correspond with the nth random fraction that is being
calculated. Remember that the series of 5 keys are used in sequence and then repeat so that random
fractions #1, #6 and #11 generated with any seed, all use key 21. So each starting number can be
matched with its own key.

I first experimented by creating a TI Multiplan spreadsheet to generate a block of random fractions using
the above formula with a series of starting numbers. I subsequently transferred this TI Multiplan
spreadsheet to MS Excel because Excel calculates faster. 

Since I had to prove that this method of
generating random fractions actually
worked, I decided to write an Extended
Basic program (named RANDOM) that
would generate some random fractions
using the above formula — but without
using RANDOMIZE and RND — and
then generate the same random
fractions using RANDOMIZE and
RND. The program only displays lists
of 16 random fractions in order to allow
some space on the screen for
descriptions of the options chosen.
Accordingly, only 16 DATA statements
have been included to store the 16
starting numbers and their related
keys. The program contains two menu
items under two options being one for
generating random fractions using
RANDOMIZE and RND and another for generating without using RANDOMIZE and RND. The two
menu items under each option do the following:

1. Generate two adjacent random
fractions (between 1 to 16) for any
of the seeds from 0 to 15.

2. Generate 16 random fractions for
any one of the seeds from 0 to 15.

The program can also print the results
of the menu choices individually or all
together. These printing options are
selected by entering P from the main
menu. Entering 1 to 4 will access the
selected option and entering E will end
the program. The printer code is set to
print to Fred Kaal’s TI99PRINT PC
program using a serial cable. This code
can be changed to suit your printer.

I also wrote a shorter program (named
RNDGEN) that will generate up to 99 random fractions for any RANDOMIZE seed from 0 to 99 using
an adaptation of the formula described above3.

In case you are wondering what prompted me to look into these RANDOMIZE peculiarities, it was a User
Note in the August 1995 issue of MICROpendium4 contained the following formula:

X(N+1) = 14389820420821*X(N)+21132486540519(MOD 10^14)

where X*() is an integer. 

I could not understand this formula but recognized that the second 14-digit number resembles the first
random fraction generated after using RANDOMIZE 0. This led to identifying other starting numbers
and observing the differences in keys described above. I still do not know what is the significance of the
first 14-digit number in this formula. None of the above observations deals with how random numbers
are generated, but only presents an observed peculiarity when RANDOMIZE is used with a seed with TI
Extended Basic.

1.1. Listing of RANDOM

100 ! SAVE DSK1.RANDOM
110 DIM SD(17),SY(17)
120 M$="M1234PEnd" :: PR$="R
S232.BA=9600" :: KY$="1" :: 
Y$="0"
130 IMAGE ### .########## .#
#########
140 IMAGE ### ### ##.#######
###
150 IMAGE ### #.##########
160 CALL CLEAR :: CALL BLUE 
:: GOSUB 5000 :: CALL PAGE1
170 CALL TITLE3 :: CALL MENU
180 DISPLAY AT(24,3):"Your C
hoice? (";M$;"):M" :: ACCEPT
 AT(24,28)SIZE(-1)VALIDATE("
M1234PE")BEEP:M1$ :: N=POS(M
$,M1$,1)
190 ON N GOSUB 170,3000,4000
,6000,7000,9000,990
200 GOTO 170
990 END
1000 DATA 21,21132486540519
1010 DATA 41,06367534286618
1020 DATA 61,74707275413897
1030 DATA 81,14807978089956
1040 DATA 01,39928199914395
1050 DATA 21,05607562158814
1060 DATA 41,45921200806813
1070 DATA 61,20620870393992
1080 DATA 81,90393996647951
1090 DATA 01,31454693928290
1100 DATA 21,02893883466609
1110 DATA 41,07587168406508
1120 DATA 61,97877041643587
1130 DATA 81,56758322465446
1140 DATA 01,39310437991685
1150 DATA 21,87605285413904
3000 DISPLAY AT(1,1)ERASE AL
L:"MENU #1 RANDOM # (1-15)?:
"&KY$ :: ACCEPT AT(1,26)VALI
DATE(DIGIT)SIZE(-2):KY$ :: I
F KY$="" THEN 3000 ELSE KY=V
AL(KY$)
3010 IF KY>15 OR KY<1 THEN 3
000
3020 DISPLAY AT(2,1):"Random
 Numbers";KY;"and";KY+1;"in 

all the Series from 0 to 15.
 Note first 2 digits."
3030 DISPLAY AT(5,1):"Uses K
eys";SY(KY-1);"and";SY(KY):"
Seed   Rnd #";KY;"    #";KY+
1
3040 IF T AND 1 THEN OPEN #1
:PR$ :: PRINT #1:"Menu #1 RA
NDOMIZE Series 0 to 15":"See
d            RND #"&STR$(KY)
&"         RND #"&STR$(KY+1)
3050 FOR X=0 TO 15
3060 A=(SY(KY-1)*10^12*X+SD(
KY-1))/10^14
3070 A1=(SY(KY)*10^12*X+SD(K
Y))/10^14
3080 IF A>1 THEN A=A-1 :: GO
TO 3080
3090 IF A1>1 THEN A1=A1-1 ::
 GOTO 3090
3100 DISPLAY AT(X+7,1):USING
 130:X,A,A1
3110 IF T AND 1 THEN PRINT #
1:X,A,A1
3120 NEXT X :: CALL ANYKEY :
: IF T AND 1 THEN CLOSE #1
3130 IF T AND 16 THEN 6000 E
LSE RETURN
4000 DISPLAY AT(1,1)ERASE AL
L:"MENU #2":"RANDOMIZE Seed 
(0-16)?:"&Y$ :: ACCEPT AT(2,
24)VALIDATE(DIGIT)SIZE(-2):Y
$
4010 IF Y$="" THEN 4000 ELSE
 Y=VAL(Y$)
4020 IF Y>16 OR Y<0 THEN 400
0
4030 DISPLAY AT(3,1):"These 
are the 1st 16 random number
s generated if seed";Y;"is u
sed.":"  #  Key  Random Numb
ers"
4040 IF T AND 2 THEN OPEN #1
:PR$ :: PRINT #1:" Menu #2 R
ANDOMIZE "&STR$(Y):" #      
       Key #         Random 
#"
4050 FOR X=0 TO 15

4060 B=(SY(X)*10^12*Y+SD(X))
/10^14
4070 IF B>1 THEN B=B-1 :: GO
TO 4070
4080 DISPLAY AT(X+7,1):USING
 140:X+1,SY(X),B
4090 IF T AND 2 THEN PRINT #
1:X+1,SY(X),B
4100 NEXT X :: CALL ANYKEY :
: IF T AND 2 THEN CLOSE #1
4110 IF T AND 16 THEN 7000 E
LSE RETURN
5000 FOR X=0 TO 15
5010 READ SY(X),SD(X)
5020 NEXT X :: RETURN
6000 DISPLAY AT(1,1)ERASE AL
L:"MENU #3 RANDOM # (1-15)?:
"&KY$ :: ACCEPT AT(1,26)VALI
DATE(DIGIT)SIZE(-2):KY$
6010 IF KY$="" THEN 6000 ELS
E KY=VAL(KY$)
6020 IF KY>15 OR KY<1 THEN 6
000
6030 DISPLAY AT(2,1):"Random
 Numbers";KY;"and";KY+1;"in 
all the Series from 0 to 15.
 Note first 2 digits."
6040 DISPLAY AT(6,1):"Seed  
 Rnd #";KY;"    #";KY+1
6050 IF T AND 4 THEN OPEN #1
:PR$ :: PRINT #1:"Menu #3 RA
NDOMIZE Series 0 to 15":"See
d            RND #"&STR$(KY)
&"         RND #"&STR$(KY+1)
6060 FOR X=0 TO 15
6070 RANDOMIZE X
6080 FOR Y=1 TO KY-1 :: K=RN
D:: NEXT Y
6090 A=RND:: B=RND:: DISPLAY
 AT(X+7,1):USING 130:X,A,B
6100 IF T AND 4 THEN PRINT #
1:X,A,B
6110 NEXT X :: CALL ANYKEY :
: IF T AND 4 THEN CLOSE #1
6120 IF T AND 16 THEN 4000 E
LSE RETURN
7000 DISPLAY AT(1,1)ERASE AL
L:"MENU #4":"RANDOMIZE Seed 
(0-16)?:"&Y$ :: ACCEPT AT(2,
24)VALIDATE(DIGIT)SIZE(-2):Y

$
7010 IF Y$="" THEN 7000 ELSE
 Y=VAL(Y$)
7020 IF Y>16 OR Y<0 THEN 700
0
7030 DISPLAY AT(3,1):"These 
are the 1st 16 random number
s generated if seed";Y;"is u
sed.":"  #  Random Numbers"
7040 IF T AND 8 THEN OPEN #1
:PR$ :: PRINT #1:"Menu #4 RA
NDOMIZE "&Y$:" #            
   Random #"
7050 RANDOMIZE Y
7060 FOR X=0 TO 15
7070 A=RND:: DISPLAY AT(X+7,
1):USING 150:X+1,A
7080 IF T AND 8 THEN PRINT #
1:X+1,A
7090 NEXT X :: CALL ANYKEY :
: IF T AND 8 THEN CLOSE #1
7100 RETURN
9000 ! SET PRINTER
9010 DISPLAY AT(1,8)ERASE AL
L:"Print Set-Up": :"Printer 
>Change as necessary":PR$ ::
 ACCEPT AT(4,1)SIZE(-28)VALI
DATE(UALPHA,DIGIT,".")BEEP:P
R$
9020 DISPLAY AT(20,1):"If pr
int 'All' pages is     selec
ted, program will run  and p
rint all pages in the  follo
wing  order 1 3 2 4."
9030 DISPLAY AT(6,1):"Set Pa
ges To Print"
9040 DISPLAY AT(8,1):"Clear 
(Y/N)"
9050 ACCEPT AT(8,14)VALIDATE
("YN")SIZE(-1):Z$ :: IF Z$="
Y" THEN T=0
9060 IF T AND 16 THEN DISPLA
Y AT(9,3):"All (Y/N)  Y" ELS
E DISPLAY AT(9,3):"All (Y/N)
"
9070 ACCEPT AT(9,14)VALIDATE
("YN")SIZE(-1):Z$ :: IF Z$="
Y" THEN T=T OR 31
9080 IF T AND 1 THEN DISPLAY
 AT(10,5):"1 (Y/N)  Y" ELSE 

DISPLAY AT(10,5):"1 (Y/N)"
9090 IF T AND 2 THEN DISPLAY
 AT(11,5):"2 (Y/N)  Y" ELSE 
DISPLAY AT(11,5):"2 (Y/N)"
9100 IF T AND 4 THEN DISPLAY
 AT(12,5):"3 (Y/N)  Y" ELSE 
DISPLAY AT(12,5):"3 (Y/N)"
9110 IF T AND 8 THEN DISPLAY
 AT(13,5):"4 (Y/N)  Y" ELSE 
DISPLAY AT(13,5):"4 (Y/N)"
9120 ACCEPT AT(10,14)VALIDAT
E("YN")SIZE(-1):Z$ :: IF Z$=
"Y" THEN T=T OR 1
9130 ACCEPT AT(11,14)VALIDAT
E("YN")SIZE(-1):Z$ :: IF Z$=
"Y" THEN T=T OR 2
9140 ACCEPT AT(12,14)VALIDAT
E("YN")SIZE(-1):Z$ :: IF Z$=
"Y" THEN T=T OR 4
9150 ACCEPT AT(13,14)VALIDAT
E("YN")SIZE(-1):Z$ :: IF Z$=
"Y" THEN T=T OR 8
9160 DISPLAY AT(17,4):"Chang
e pages? (Y/N) N" :: ACCEPT 
AT(17,24)VALIDATE("YN")SIZE(
-1)BEEP:AN$ :: IF AN$="Y" TH
EN 9030
9170 IF T AND 16 THEN 3000 E
LSE RETURN
10000 SUB ANYKEY
10010 DISPLAY AT(24,2)BEEP:"
Press Any Key to Continue"
10020 CALL KEY(3,K,S):: IF S
<1 THEN 10020
10030 SUBEND
10100 SUB MENU
10110 DISPLAY AT(5,1):"Witho
ut TI RANDOMIZE and RND":"1 
Display Adjacent Random     
Numbers for a Series of     
16 RANDOMIZE Seeds"
10120 DISPLAY AT(10,1):"2 Di
splay First 16 Random     Nu
mbers for a Specific      Se
ed"

10130 DISPLAY AT(14,1):"Usin
g TI RANDOMIZE and RND":"3 D
isplay Adjacent Random     N
umbers for a Series of     1
6 RANDOMIZE Seeds"
10140 DISPLAY AT(19,1):"4 Di
splay First 16 Random     Nu
mbers for a Specific      Se
ed"
10150 DISPLAY AT(23,1):"M Th
is Menu  P Set Printer"
10160 SUBEND
11000 SUB PAGE1
11010 CALL TITLE3
11020 DISPLAY AT(5,1):"This 
program explores some  pecul
arities in the use of  the R
ANDOMIZE statement and will 
create series of random"
11030 DISPLAY AT(9,1):"numbe
rs without using the   RANDO
MIZE statement or RND  funct
ion. For comparison    the s
ame random numbers can"
11040 DISPLAY AT(13,1):"be c
reated using RANDOMIZE  and 
RND. Output can also be sent
 to a printer."
11090 CALL ANYKEY :: SUBEND
29245 SUB TITLE3
29250 ! SHORT TITLE SCREEN
29255 DISPLAY AT(1,8)ERASE A
LL:"Random Numbers" :: CALL 
CHAR(95,"00FF"):: CALL HCHAR
(2,10,95,14)
29260 DISPLAY AT(3,6):"By Ja
cques Groslouis" :: CALL HCH
AR(4,8,95,20)
29265 SUBEND
29505 SUB BLUE
29510 ! SWITCHES DISPLAY TO 
WHITE ON BLUE; JLS 7/88
29515 CALL SCREEN(5):: FOR L
=0 TO 14 :: CALL COLOR(L,16,
1):: NEXT L :: SUBEND


